

NAG Library Routine Document

G08AHF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G08AHF performs the Mann–Whitney U test on two independent samples of possibly unequal size.

2 Specification

SUBROUTINE G08AHF (N1, X, N2, Y, TAIL, U, UNOR, P, TIES, RANKS, WRK, IFAIL)

INTEGER N1, N2, IFAIL
 REAL (KIND=nag_wp) X(N1), Y(N2), U, UNOR, P, RANKS(N1+N2), WRK(N1+N2)
 LOGICAL TIES
 CHARACTER(1) TAIL

3 Description

The Mann–Whitney U test investigates the difference between two populations defined by the distribution functions $F(x)$ and $G(y)$ respectively. The data consist of two independent samples of size n_1 and n_2 , denoted by x_1, x_2, \dots, x_{n_1} and y_1, y_2, \dots, y_{n_2} , taken from the two populations.

The hypothesis under test, H_0 , often called the null hypothesis, is that the two distributions are the same, that is $F(x) = G(x)$, and this is to be tested against an alternative hypothesis H_1 which is

$H_1: F(x) \neq G(y)$; or

$H_1: F(x) < G(y)$, i.e., the x 's tend to be greater than the y 's; or

$H_1: F(x) > G(y)$, i.e., the x 's tend to be less than the y 's,

using a two tailed, upper tailed or lower tailed probability respectively. You select the alternative hypothesis by choosing the appropriate tail probability to be computed (see the description of parameter TAIL in Section 5).

Note that when using this test to test for differences in the distributions one is primarily detecting differences in the location of the two distributions. That is to say, if we reject the null hypothesis H_0 in favour of the alternative hypothesis $H_1: F(x) > G(y)$ we have evidence to suggest that the location, of the distribution defined by $F(x)$, is less than the location, of the distribution defined by $G(y)$.

The Mann–Whitney U test differs from the Median test (see G08ACF) in that the ranking of the individual scores within the pooled sample is taken into account, rather than simply the position of a score relative to the median of the pooled sample. It is therefore a more powerful test if score differences are meaningful.

The test procedure involves ranking the pooled sample, average ranks being used for ties. Let r_{1i} be the rank assigned to x_i , $i = 1, 2, \dots, n_1$ and r_{2j} the rank assigned to y_j , $j = 1, 2, \dots, n_2$. Then the test statistic U is defined as follows;

$$U = \sum_{i=1}^{n_1} r_{1i} - \frac{n_1(n_1 + 1)}{2}$$

U is also the number of times a score in the second sample precedes a score in the first sample (where we only count a half if a score in the second sample actually equals a score in the first sample).

G08AHF returns:

- (a) The test statistic U .
 (b) The approximate Normal test statistic,

$$z = \frac{U - \text{mean}(U) \pm \frac{1}{2}}{\sqrt{\text{var}(U)}}$$

where

$$\text{mean}(U) = \frac{n_1 n_2}{2}$$

and

$$\text{var}(U) = \frac{n_1 n_2 (n_1 + n_2 + 1)}{12} - \frac{n_1 n_2}{(n_1 + n_2)(n_1 + n_2 - 1)} \times TS$$

where

$$TS = \sum_{j=1}^{\tau} \frac{(t_j)(t_j - 1)(t_j + 1)}{12}$$

τ is the number of groups of ties in the sample and t_j is the number of ties in the j th group.

Note that if no ties are present the variance of U reduces to $\frac{n_1 n_2}{12}(n_1 + n_2 + 1)$.

- (c) An indicator as to whether ties were present in the pooled sample or not.
 (d) The tail probability, p , corresponding to U (adjusted to allow the complement to be used in an upper one tailed or a two tailed test), depending on the choice of TAIL, i.e., the choice of alternative hypothesis, H_1 . The tail probability returned is an approximation of p is based on an approximate Normal statistic corrected for continuity according to the tail specified. If n_1 and n_2 are not very large an exact probability may be desired. For the calculation of the exact probability see G08AJF (no ties in the pooled sample) or G08AKF (ties in the pooled sample).

The value of p can be used to perform a significance test on the null hypothesis H_0 against the alternative hypothesis H_1 . Let α be the size of the significance test (that is, α is the probability of rejecting H_0 when H_0 is true). If $p < \alpha$ then the null hypothesis is rejected. Typically α might be 0.05 or 0.01.

4 References

Conover W J (1980) *Practical Nonparametric Statistics* Wiley

Neumann N (1988) Some procedures for calculating the distributions of elementary nonparametric teststatistics *Statistical Software Newsletter* **14(3)** 120–126

Siegel S (1956) *Non-parametric Statistics for the Behavioral Sciences* McGraw–Hill

5 Parameters

- 1: N1 – INTEGER *Input*
On entry: the size of the first sample, n_1 .
Constraint: $N1 \geq 1$.
- 2: X(N1) – REAL (KIND=nag_wp) array *Input*
On entry: the first vector of observations, x_1, x_2, \dots, x_{n_1} .
- 3: N2 – INTEGER *Input*
On entry: the size of the second sample, n_2 .
Constraint: $N2 \geq 1$.

- 4: Y(N2) – REAL (KIND=nag_wp) array Input
On entry: the second vector of observations. y_1, y_2, \dots, y_{n_2} .
- 5: TAIL – CHARACTER(1) Input
On entry: indicates the choice of tail probability, and hence the alternative hypothesis.
 TAIL = 'T'
 A two tailed probability is calculated and the alternative hypothesis is $H_1 : F(x) \neq G(y)$.
 TAIL = 'U'
 An upper tailed probability is calculated and the alternative hypothesis $H_1 : F(x) < G(y)$, i.e., the x 's tend to be greater than the y 's.
 TAIL = 'L'
 A lower tailed probability is calculated and the alternative hypothesis $H_1 : F(x) > G(y)$, i.e., the x 's tend to be less than the y 's.
Constraint: TAIL = 'T', 'U' or 'L'.
- 6: U – REAL (KIND=nag_wp) Output
On exit: the Mann–Whitney rank sum statistic, U .
- 7: UNOR – REAL (KIND=nag_wp) Output
On exit: the approximate Normal test statistic, z , as described in Section 3.
- 8: P – REAL (KIND=nag_wp) Output
On exit: the tail probability, p , as specified by the parameter TAIL.
- 9: TIES – LOGICAL Output
On exit: indicates whether the pooled sample contained ties or not. This will be useful in checking which routine to use should one wish to calculate an exact tail probability.
 TIES = .FALSE., no ties were present (use G08AJF for an exact probability).
 TIES = .TRUE., ties were present (use G08AKF for an exact probability).
- 10: RANKS(N1 + N2) – REAL (KIND=nag_wp) array Output
On exit: contains the ranks of the pooled sample. The ranks of the first sample are contained in the first N1 elements and those of the second sample are contained in the next N2 elements.
- 11: WRK(N1 + N2) – REAL (KIND=nag_wp) array Workspace
- 12: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by $X04AAF$).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $N1 < 1$,
or $N2 < 1$.

$IFAIL = 2$

On entry, $TAIL \neq 'T', 'U'$ or $'L'$.

$IFAIL = 3$

The pooled sample values are all the same, that is the variance of $U = 0.0$.

7 Accuracy

The approximate tail probability, p , returned by G08AHF is a good approximation to the exact probability for cases where $\max(n_1, n_2) \geq 30$ and $(n_1 + n_2) \geq 40$. The relative error of the approximation should be less than 10%, for most cases falling in this range.

8 Further Comments

The time taken by G08AHF increases with n_1 and n_2 .

9 Example

This example performs the Mann–Whitney test on two independent samples of sizes 16 and 23 respectively. This is used to test the null hypothesis that the distributions of the two populations from which the samples were taken are the same against the alternative hypothesis that the distributions are different. The test statistic, the approximate Normal statistic and the approximate two tail probability are printed. An exact tail probability is also calculated and printed depending on whether ties were found in the pooled sample or not.

9.1 Program Text

```

Program g08ahfe
!      G08AHF Example Program Text
!      Mark 24 Release. NAG Copyright 2012.
!
!      .. Use Statements ..
Use nag_library, Only: g08ahf, g08ajf, g08akf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: p, pexact, u, unor
Integer                    :: ifail, liwrk, lwrk, lwrk1, lwrk2,    &
                             lwrk3, mn, n1, n2, nsum
Logical                    :: ties
Character (1)              :: tail
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: ranks(:), wrk(:), x(:), y(:)
Integer, Allocatable        :: iwrk(:)
!      .. Intrinsic Procedures ..
Intrinsic                   :: int, max, min

```

```

!      .. Executable Statements ..
      Write (nout,*) 'G08AHF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Read in problem size
      Read (nin,*) n1, n2, tail

!      Calculate sizes of various workspaces
      nsum = n1 + n2
      mn = min(n1,n2)

!      Workspace for G08AHF
      lwrk1 = nsum

!      Workspace for G08AJF
      lwrk2 = int(n1*n2/2) + 1

!      Workspace for G08AKF
      lwrk3 = mn + mn*(mn+1)*nsum - mn*(mn+1)*(2*mn+1)/3 + 1
      liwrk = 2*nsum + 2

      lwrk = max(lwrk1,lwrk2,lwrk3)
      Allocate (x(n1),y(n2),ranks(nsum),wrk(lwrk),iwrk(liwrk))

!      Read in data
      Read (nin,*) x(1:n1)
      Read (nin,*) y(1:n2)

!      Display title
      Write (nout,*) 'Mann-Whitney U test'
      Write (nout,*)

!      Display input data
      Write (nout,99999) 'Sample size of group 1 = ', n1
      Write (nout,99999) 'Sample size of group 2 = ', n2
      Write (nout,*)
      Write (nout,*) 'Data values'
      Write (nout,*)
      Write (nout,99998) '   Group 1   ', x(1:n1)
      Write (nout,*)
      Write (nout,99998) '   Group 2   ', y(1:n2)

!      Perform test
      ifail = 0
      Call g08ahf(n1,x,n2,y,tail,u,unor,p,ties,ranks,wrk,ifail)

!      Calculate exact probabilities
      If (.Not. ties) Then
         ifail = 0
         Call g08ajf(n1,n2,tail,u,pexact,wrk,lwrk,ifail)

      Else
         ifail = 0
         Call g08akf(n1,n2,tail,ranks,u,pexact,wrk,lwrk,iwrk,ifail)
      End If

!      Display results
      Write (nout,*)
      Write (nout,99997) 'Test statistic           = ', u
      Write (nout,99997) 'Normal Statistic           = ', unor
      Write (nout,99997) 'Approx. tail probability = ', p
      Write (nout,*)
      If (ties) Then
         Write (nout,*) 'There are ties in the pooled sample'
      Else
         Write (nout,*) 'There are no ties in the pooled sample'
      End If
      Write (nout,*)

```

```

      Write (nout,99997) 'Exact tail probability = ', pexact
99999 Format (1X,A,I5)
99998 Format (1X,A,8F5.1,2(/14X,8F5.1))
99997 Format (1X,A,F10.4)
      End Program g08ahfe

```

9.2 Program Data

```

G08AHF Example Program Data
 16 23 'L'                                :: N1,N2,TAIL
 13.0 6.0 12.0 7.0 12.0 7.0 10.0 7.0
 10.0 7.0 16.0 7.0 10.0 8.0 9.0 8.0    :: End of X
 17.0 6.0 10.0 8.0 15.0 8.0 15.0 10.0
 15.0 10.0 14.0 10.0 14.0 11.0 14.0 11.0
 13.0 12.0 13.0 12.0 13.0 12.0 12.0    :: End if Y

```

9.3 Program Results

G08AHF Example Program Results

Mann-Whitney U test

Sample size of group 1 = 16
 Sample size of group 2 = 23

Data values

```

Group 1  13.0  6.0 12.0  7.0 12.0  7.0 10.0  7.0
         10.0  7.0 16.0  7.0 10.0  8.0  9.0  8.0

```

```

Group 2  17.0  6.0 10.0  8.0 15.0  8.0 15.0 10.0
         15.0 10.0 14.0 10.0 14.0 11.0 14.0 11.0
         13.0 12.0 13.0 12.0 13.0 12.0 12.0

```

Test statistic = 86.0000
 Normal Statistic = -2.8039
 Approx. tail probability = 0.0025

There are ties in the pooled sample

Exact tail probability = 0.0020
